
/* Filename:   GLS2.LIM                                        */
/* Date:       30 June 1998                                    */
/* Project:    Determinants of Youth Employment                */
/* Written by: Owen Gabbitas (Trade & Economic Studies Branch) */

/* Purpose:    Conducts SURE regressions using GLS             */
/*             with aggregated youth                           */

Open; output=v:\youthemp\time\limdep\gls2.out $
Title; output file v:\..\gls2.out (aggregated youth) $

Reset $

/* ==== Read in data - variable names in first line ==== */

Read; file = v:\youthemp\time\limdep\input2.wk1
    ; format = wks
    ; names = $

/* y - youth  (aged 15 to 19) */
/* a - adults (aged 20 to 64) */
/* m - male                   */
/* f - female                 */
/* ie. yf - female youth      */

/* list; Cy, Wy, Edy, My $ */
/* list; Cam, Wam, Edam, Mam $ */
/* list; Caf, Waf, Edaf, Maf $ */
/* list; Ck, r $ */
/* list; Year, Industry, Q $ */

Create; LWy=log(Wy)
      ; LWam=log(Wam)
      ; LWaf=log(Waf)
 ; LWk=log(r)
 ; LQ = log (Q) $

Namelist ; Wages = LWy, LWam, LWaf
    ; Prices = LWy, LWam, LWaf, LWk
    ; Costshar = Cy, Cam, Caf
    ; Ed = Edy, Edam, Edaf $

/* ==== Seemingly unrelated regressions (SURE) - GLS ==== */

/* Unconstrained */
Sure; LHS = Cy, Cam, Caf
    ; Eq1 = one, Prices, LQ
    ; Eq2 = one, Prices, LQ
    ; Eq3 = one, Prices, LQ $

/* Imposing Symmetry only */
Sure; LHS = Cy, Cam, Caf
    ; Eq1 = one, Prices, LQ
    ; Eq2 = one, Prices, LQ
    ; Eq3 = one, Prices, LQ
    ; Cls:     B(3)  - B(8)  = 0,
     B(4)  - B(14) = 0,
     B(10) - B(15) = 0 $



/* Imposing homogeneity only */
Sure; LHS = Cy, Cam, Caf
    ; Eq1 = one, Prices, LQ
    ; Eq2 = one, Prices, LQ
    ; Eq3 = one, Prices, LQ
    ; Cls:     B(2)  + B(9)  + B(16) = 1,
     B(2)  + B(3)  + B(4)   + B(5)  = 0,
     B(8)  + B(9)  + B(10)  + B(11) = 0,
     B(14) + B(15) + B(16)  + B(17) = 0,
     B(1)  + B(7)  + B(13) = 0 $

/* Imposing both symmetry and homogeneity */
Sure; LHS = Cy, Cam, Caf
    ; Eq1 = one, Prices, LQ
    ; Eq2 = one, Prices, LQ
    ; Eq3 = one, Prices, LQ
    ; Cls:     B(3)  - B(8)         = 0,
     B(4)  - B(14)                  = 0,
     B(10) - B(15)                  = 0,
     B(2)  + B(9)  + B(16) = 1,
     B(2)  + B(3)  + B(4)   + B(5)  = 0,
     B(8)  + B(9)  + B(10)  + B(11) = 0,
     B(14) + B(15) + B(16)  + B(17) = 0,
     B(1)  + B(6)  + B(11)          = 0 $


